

Chapter 3

Data Sets and Parametric Curves

The chapter discusses ways to describe mathematical quantities other than expressions, functions and equations. It begins with the Maple commands to plot data points and fit a polynomial curve to a given set of data both exactly and approximately. Then it describes how to plot parametric curves, with applications to inverse functions and polar curves.

This chapter can be safely skipped until data sets, parametric curves, inverse functions or polar curves are needed.

3.1 Data Points and Curve Fitting


Often, problems involve data points rather than functions or expressions or equations. You may want to plot these points, and you may want to find a curve which passes through these points, either exactly or approximately.

Data Sets: In Maple, an ordered list must be enclosed in square brackets `[]`. So `[1,2.2]` is an ordered pair, or point, while `[[1,2.2], [3,5.5], [6,4.2]]` is an ordered list of points. To plot this list of points, first assign it to a variable, say *mydata*

```
> mydata:=[[1,2.2], [3,5.5], [6,4.2]];
      mydata := [[1, 2.2], [3, 5.5], [6, 4.2]]
```

and check for typing errors. Then issue the `plot` command


```
> plot(mydata, style=point);
```



NOTE: You have already seen point plots using the `style=point` option in Section 1.4 when we discussed plots of discontinuous functions.


You can change the x - and y -ranges for the plot, either explicitly (as the second and third options) or by including a `view` option. You can change the symbol used for the points and the size of the symbol by using the `symbol` and `symbolsize` options. For example:

```
> plot(mydata,0..7,0..6, style=point, symbol=cross,
> symbolsize=24);
```



Deleting the option `style=point` will draw this data set with connecting line segments.

```
> plot(mydata, view=[0..7,0..6]);
```



Exact Curve Fitting: Looking at the plot, or for other physical reasons, you might want to connect the dots with a parabola instead of line segments.

EXAMPLE 1: Find the expression whose graph is the parabola that passes through the points (1, 2.2), (3, 5.5), and (6, 4.2). Then graph the parabola and the points in the same plot.

SOLUTION: First, enter the general formula for a parabola (as a function of x).

```
> p:=x->a*x^2+b*x+c;
```

$$p := x \rightarrow ax^2 + bx + c$$

The unknown coefficients a , b , and c must be found so that the parabola passes through the points (1, 2.2), (3, 5.5), and (6, 4.2). In order for the parabola to pass through the point (1, 2.2), the equation $p(1) = 2.2$ must be satisfied. We enter this equation and label it *eq1*:

```
> eq1:=p(1)=2.2;
```

$$eq1 := a + b + c = 2.2$$

Similarly, the other two equations are.

```
> eq2:=p(3)=5.5; eq3:=p(6)=4.2;
```

$$eq2 := 9a + 3b + c = 5.5$$

$$eq3 := 36a + 6b + c = 4.2$$

These equations are then solved and the solution is labeled *sol*.

```
> sol:=solve({eq1,eq2,eq3},{a,b,c});
```

$$sol := \{a = -0.4166666667, b = 3.3166666667, c = -0.7000000000\}$$

(The order of the variables in your solution may differ.) These values of a , b , and c can be substituted into $p(x)$ using the `subs` command.

```
> q:=subs(sol,p(x));
```

$$q := -0.4166666667x^2 + 3.3166666667x - 0.7000000000$$

Here the label q is given to the parabola. Note that q is defined as a *Maple expression*.

NOTE: As mentioned in Sections 1.4 and 2.2, two or more expressions (or functions) can be plotted on the same interval by enclosing the expressions (or functions) in curly braces $\{ \}$. However, if the variables or intervals are different, or if they are two different kinds of plots (expressions vs. functions vs. equations vs. points) then you must use the `display` command from the `plots` package.

To plot the parabola and the points, we first plot them separately and label the plots:


```
> p1:=plot(mydata,0..7,0..6, style=point, symbol=cross,
```

```
> symbolsize=36):
```

```
> p2:=plot(q,x=0..7):
```

Notice we ended these commands with colons (rather than semicolons) to suppress output. Next we load the `plots` package and `display` the two plots (this time with a semicolon).

```
> with(plots): display([p1,p2]);
```



The points are right on the parabola.

Approximate Curve Fitting: If there are many more points, you may want to find a curve which best fits the data but does not necessarily go precisely through the points. This is simple using Maple's `stats` and `statplots` packages.

EXAMPLE 2: Consider the data points given in the table below:

x	1	2	3	4	5	6	7
y	2.2	3.7	5.5	5.9	5.6	4.2	3.1


Find a parabola which best fits the data. Then graph the parabola and the points in the same plot.

SOLUTION: Load the `stats` and `statplots` packages and define lists of the x - and y -coordinates in the data.

```
> with(stats): with(statplots):
> xs:= [1,2,3,4,5,6,7]; ys:= [2.2,3.7,5.5,5.9,5.6,4.2,3.1];
      xs := [1, 2, 3, 4, 5, 6, 7]
      ys := [2.2, 3.7, 5.5, 5.9, 5.6, 4.2, 3.1]
```

Plot the data using the `scatterplot` command from the `statplots` package.

```
> p1:=scatterplot(xs,ys, symbol=cross, symbolsize=24): p1;
```




Although this method of plotting points might seem more complicated, it has some advantages. For example, Maple's built-in curve fitting commands require the x - and y -coordinates to be entered separately.

From the plot, it appears as though a quadratic function might give a good approximation to this data. The `fit[leastsquare[...]]` command from the `stats` package will create a quadratic least squares fit.

```
> fit[leastsquare][x,y],y=a*x^2+b*x+c]([xs,ys]);
      y = -0.3619047619 x^2 + 3.030952381 x - 0.5714285714
> q:=rhs(%);
      q := -0.3619047619 x^2 + 3.030952381 x - 0.5714285714
```

Let's see how well we did.

```
> p2:=plot(q,x=0..8):
> display({p1,p2});
```



Not bad.

The `fit[leastsquare[...]]` command can be easily modified to fit other types of curves.

EXAMPLE 3: Consider the population data given in the table below.

Date	1900	1910	1920	1930	1940	1950
Pop. (billions)	1.65	1.75	1.86	2.07	2.30	2.52
	1960	1970	1980	1990	1996	
	3.02	3.70	4.45	5.30	5.77	

Enter this data into Maple, obtain a cubic fit and plot the data with the cubic function.


SOLUTION: The time values are in decades since 1900.

```
> ts:=[0,1,2,3,4,5,6,7,8,9,9.6]:
> Ps:=[1.65,1.75,1.86,2.07,2.30,2.52,3.02,3.70,4.45,5.30,5.77]:
> fit[leastsquare][[t,P],P=a*t^3+b*t^2+c*t+d]([ts,Ps]);
P = 0.002325670340 t^3 + 0.01914440256 t^2 + 0.03583942511 t + 1.676870710
> f:=rhs(%);
f := 0.002325670340 t^3 + 0.01914440256 t^2 + 0.03583942511 t + 1.676870710
```

```

> p1:=scatterplot(ts, Ps, symbol=cross, symbolsize=24):
> p2:=plot(f,t=0..10):
> display({p1,p2});

```



Again a good fit.

3.2 Parametric Curves

Some curves, such as circles and ellipses, are not the graphs of functions. Other curves are not even conveniently given by equations. Instead, these figures are more conveniently described by *parametric equations*, which are of the form

$$x = f(t), \quad y = g(t), \quad \text{for } a \leq t \leq b$$

where f and g are functions of the parameter t , and a and b are the initial and final values of t . If you want, you can think of the equation $(x, y) = (f(t), g(t))$ as giving the position (x, y) of a particle in a plane as a function of time t . However, in general, t may not represent time (and you do not need to use the letter t for the parameter). It may represent some other physical or geometrical quantity such as angle or arclength.

To plot the parametric curve $x = f(t)$, $y = g(t)$, define the component functions as Maple expressions \mathbf{f} and \mathbf{g} and type the command `plot([f,g,t=a..b])`. NOTE: It is easy to confuse the syntax for a parametric plot with the square bracket syntax for plotting two expressions at the same time. For parametric plots, square brackets `[]` must be used and the parameter and its range must be *inside* the brackets. To plot two functions, you may use square brackets `[]` but the variable and range must be *outside* the brackets.


EXAMPLE 1: Plot the circle of radius 3 centered at $[2, 1]$.

SOLUTION: The circle may be parametrized by:

$$x = 2 + 3 \cos(\theta) \quad y = 1 + 3 \sin(\theta), \quad \text{for } 0 \leq \theta \leq 2\pi$$

where the parameter is the angle θ . To plot the circle we enter the coordinates as Maple expressions and plot it:

```
> f,g:=2+3*cos(theta),1+3*sin(theta);
      f, g := 2 + 3 cos(theta), 1 + 3 sin(theta)
> plot([f,g,theta=0..2*Pi], scaling=constrained);
```



NOTE: We have included the `scaling=constrained` option to equalize the scales on the axes.


EXAMPLE 2: *Lissajous figures* A Lissajous figure is a curve in the plane which may be parametrized as

$$(x, y) = (\cos(pt), \sin(qt))$$

where p and q are positive integers which are relatively prime (no common factors other than 1). Plot the Lissajous figure with $p = 5$ and $q = 4$.

SOLUTION: Here it is:

```
> plot([cos(5*t),sin(4*t),t=0..2*Pi]);
```



As t varies from 0 to 2π , the x -coordinate oscillates $p = 5$ times while the y -coordinate oscillates $q = 4$ times. (There are 5 bumps on the left and right and 4 bumps on the top and bottom.)

3.3 Inverse Functions

The inverse of a function f is a function f^{-1} which undoes f . Thus

$$y = f^{-1}(x) \quad \text{means} \quad x = f(y)$$

For example, the cube root is the inverse of the cube function.

EXAMPLE 1: Find the inverse of the function $f(x) = \sqrt[5]{(x+4)^3 - 7}$. Then graph the function, the inverse and the 45° line $y = x$ in the same plot.

SOLUTION: Define the equation $y = f(x)$:

```
> eq1:=y=((x+4)^3-7)^(1/5);
```

$$eq1 := y = ((x + 4)^3 - 7)^{(1/5)}$$

Interchange x and y : (The braces make the substitutions occur simultaneously.)

```
> eq2:=subs({x=y,y=x},eq1);
```

$$eq2 := x = ((y + 4)^3 - 7)^{(1/5)}$$

Solve for y :

```
> sol:=solve(eq2,y);
```

$$sol := (7 + x^5)^{(1/3)} - 4, -\frac{(7 + x^5)^{(1/3)}}{2} + \frac{1}{2}I\sqrt{3}(7 + x^5)^{(1/3)} - 4, \\ -\frac{(7 + x^5)^{(1/3)}}{2} - \frac{1}{2}I\sqrt{3}(7 + x^5)^{(1/3)} - 4$$


We only want the real solution. So

```
> eq3:=y=sol[1];
```

$$eq3 := y = (7 + x^5)^{(1/3)} - 4$$

Now we can plot the two functions and the diagonal:

```
> plot([rhs(eq1),rhs(eq3),x], x=-2..4, y=-2..4,
> linestyle=[SOLID,DASH,DOT], thickness=[5,3,1],
> scaling=constrained);
```



Notice that the dashed inverse function is the mirror image (through the diagonal) of the original solid function. This is because the inverse function is obtained by interchanging x and y (and solving for y).

Frequently, it is impossible to solve explicitly for an inverse function. However, it is still possible to plot the inverse function using a parametric plot. Given a function $y = f(x)$, its parametric form is $x = t$, $y = f(t)$. Since the inverse function $y = f^{-1}(x)$ satisfies $x = f(y)$, its parametric form is $x = f(t)$, $y = t$.

EXAMPLE 2: Plot the function $f(x) = 4 + x + \sin(x)$ and its inverse.


NOTE: It is impossible to solve the equation $x = 4 + y + \sin(y)$ for y .

SOLUTION: We enter the function, but as an expression in t :

```
> f:=4+t+sin(t);
      f := 4 + t + sin(t)
```


Then the parametric plot of the function is

```
> plot([t,f,t=-10..10], view=[-10..10,-10..10],
> scaling=constrained);
```



and the parametric plot of the inverse function is

```
> plot([f,t,t=-10..10], view=[-10..10,-10..10],
> scaling=constrained);
```



Notice that the inverse is the reflection of the function through the diagonal.

More generally, if $x = f(t)$, $y = g(t)$ is the parametric form of a function, then $x = g(t)$, $y = f(t)$ is the parametric form of its inverse function.

Sometimes texts obfuscate the issue of inverses by placing undue emphasis on whether or not the (original or) inverse is a function, i.e., passes “the vertical


line test.” As shown below, one can *always* graph the inverse of a “relation” given by an equation; one simply reverses the variables and uses `implicitplot`. In the following example, the original curve is a function since it passes the vertical line test, but the original is not 1-1 since it does not pass the horizontal line test. So its inverse relation (with the variables flipped) will not pass a vertical line test. So the inverse curve exists, but is not a function. We still have symmetry with respect to the 45° line and both graphs are easy to draw using `implicitplot`.

EXAMPLE 3: Plot the relation $x^2 + y^3 = 1$ and its inverse and the 45° line in one plot. Give the different relations different colors and thicknesses. Notice that the original relation is a function. Then select a maximal interval on which the original function is 1-1. What is the domain of its inverse? Plot them both with the 45° line in one plot.

SOLUTION: We enter the relation, interchange the variables to get the inverse relation and plot them:


```
> eqn:=x^2+y^3=1;
> inv:=subs({x=y,y=x},eqn);
      eqn := x2 + y3 = 1
      inv := y2 + x3 = 1

> with(plots):
> implicitplot([eqn,inv,y=x],x=-3..3,y=-3..3, scaling=constrained,
> thickness=[1,3,1], color=[red,blue,black]);
```



Select the interval $[0, \infty)$. The domain of the inverse will be $(-\infty, 1]$. We plot them: (We plot using a 3 instead of ∞ .)

```
> p1:=implicitplot(eqn,x=0..3,y=-3..1, scaling=constrained,
> thickness=1, color=red): p1;
> p2:=implicitplot(inv,x=-3..1,y=0..3, scaling=constrained,
> thickness=3, color=blue): p2;
> p3:=implicitplot(y=x,x=-3..3,y=-3..3): p3;
> display([p1,p2,p3]);
```



3.4 Polar Curves

Polar coordinates (r, θ) are related to rectangular coordinates (x, y) by

$$x = r \cos \theta \quad y = r \sin \theta$$

$$r = \sqrt{x^2 + y^2} \quad \tan \theta = \frac{y}{x}$$

A polar curve is a special case of a parametric curve in which the radius r is given as a function of the angle θ which serves as the parameter. So if $r = f(\theta)$ is a polar equation, then the parametric equations are

$$x = r \cos(\theta) = f(\theta) \cos(\theta) \quad y = r \sin(\theta) = f(\theta) \sin(\theta).$$

There are three ways to plot a polar curve: as an ordinary parametric curve, by including the `coords=polar` option in a `plot` command, or by using the `polarplot` command from the `plots` package.

EXAMPLE 1: Plot the polar equation

$$r = \cos(3\theta) \quad \text{for } 0 \leq \theta \leq 2\pi$$

SOLUTION: To plot the polar equation $r = \cos(3\theta)$, convert it to parametric form as

```
> r1:=cos(3*theta);
```


```
      r1 := cos(3θ)
```

```
> x1,y1:=r1*cos(theta),r1*sin(theta);
```

```
      x1, y1 := cos(3θ) cos(θ), cos(3θ) sin(θ)
```

and plot it as a parametric curve:

```
> plot([x1,y1,theta=0..2*Pi], scaling=constrained);
```



The exact same curve is produced by the `plot` command with the `coords=polar` option:

```
> plot(r1,theta=0..2*Pi, coords=polar, scaling=constrained);
```

and by the `polarplot` command from the `plots` package:

```
> with(plots):
```

```
> polarplot(r1,theta=0..2*Pi, scaling=constrained);
```

EXAMPLE 2: Show that the polar equation $r = \frac{1}{1 - \sin \theta}$ is a parabola and plot it.

SOLUTION: Enter r as an expression and as an equation.

```
> r1:=1/(1-sin(theta)); eq1:=r=r1;
```

$$r1 := \frac{1}{1 - \sin(\theta)}$$

$$eq1 := r = \frac{1}{1 - \sin(\theta)}$$

Substitute $\sin(\theta) = \frac{y}{r}$ and $r = \sqrt{x^2 + y^2}$, and solve for y : (Notice the substitutions are done successively, not simultaneously, by leaving out the braces.)

```
> eq2:=subs(sin(theta)=y/r, r=sqrt(x^2+y^2), eq1);
```

$$eq2 := \sqrt{x^2 + y^2} = \frac{1}{1 - \frac{y}{\sqrt{x^2 + y^2}}}$$

```
> sol:=solve(eq2,y);
```

$$sol := x I, -I x, \frac{x^2}{2} - \frac{1}{2}$$


Ignore the imaginary solutions. So the equation is

```
> y=sol[3];
```

$$y = \frac{x^2}{2} - \frac{1}{2}$$


which is a parabola. Now plot it for $0 \leq \theta \leq 2\pi$.

```
> polarplot(r1,theta=0..2*Pi);
```



Notice that there are extraneous lines and the ranges are way too large due to the fact that the expression r is large for θ near $\pi/2$. To view a portion of the graph near the origin, add the view option:

```
> polarplot(r1,theta=0..2*Pi, view=[-4..4,-3..5]);
```



3.5 Summary

- Given a set of data as a list of points


```
> mydata:=[[ , ], [ , ], ..., [ , ]];
```

 plot it using


```
> plot(mydata, style=point);
```

 and be able to find a polynomial passing through these points.
- Given a set of data as two lists of x - and y -coordinates


```
> xs:=[ , , ..., ]; ys:=[ , , ..., ];
```

 plot it using


```
> with(stats): with(statplots):
      > p1:=scatterplot(xs,ys): p1;
```

 Also find the curve which best fits the data using

```
> fit[leastsquare[[x,y],y=expr]]([xs,ys]);
> q:=rhs(%);
```

and plot it with the data set using

```
> p2:=plot(q,x):
> display({p1,p2});
```

- Given a parametric curve

$$x = f(t) \quad y = g(t), \quad \text{for } a \leq t \leq b$$

plot it using

```
> plot([f,g,t=a..b], scaling=constrained);
```

where f and g are expressions in t .

- The inverse of a function f is the function f^{-1} defined by

$$y = f^{-1}(x) \quad \text{means} \quad x = f(y)$$

To find f^{-1} start with the equation

```
> eq1:=y=expr;
```

where expr is an expression in x . Interchange x and y :

```
> eq2:=subs({x=y,y=x},eq1);
```

and solve for y :

```
> eq3:=y=solve(eq2,y);
```

(It may be necessary to select the real solution.) Then the function and its inverse and the diagonal may be plotted using

```
> plot([rhs(eq1),rhs(eq3),x], x=a..b, y=c..d,
> scaling=constrained);
```

- If a parametric curve $x = f(t)$, $y = g(t)$ is the graph of a function, then the inverse function is given parametrically as $x = g(t)$, $y = f(t)$. The function and its inverse may be plotted using

```
> plot({[f,g,t=a..b],[g,f,t=a..b]}, view=[c..d,c..d],
> scaling=constrained);
```

- The polar curve

$$r = f(\theta) \quad \text{for } a \leq \theta \leq b$$

may be plotted using

```
> polarplot(f, theta=a..b, scaling=constrained);
```

where f is an expression in θ .

3.6 Exercises

- Find the cubic polynomial whose graph passes through the points (2,3.2), (3,2.6), (5,1.2), and (7,3.3). Graph the points and your answer and put them into one plot using the `display` command.
- Draw a map of Texas. To plot a map of Texas, enter the following two lists for the northern and southern boundaries of Texas.


```
> north=[[0,0], [3,0], [3,4.5], [6,4.5], [6,2.2], [7,2.1],
> [ 8,1.8], [9,1.9], [10,1.8], [11,1.7], [11,-2.2]];
> south=[[0,0], [1,-1.1], [2,-2.5], [3,-2.9], [4,-2.3],
> [5,-2.8], [6,-4.4], [7,-5.8], [8,-6.1], [9,-3.3],
> [10,-2.8], [11,-2.2]];
```

Here, the origin is the westernmost corner of Texas (near El Paso) and the x -axis is the extension of the east-west border between New Mexico and Texas. Each unit represents approximately 69 miles. After entering these lists, execute:

```
> plot({north, south});
```

- The general equation of a circle is

$$x^2 + y^2 + ax + by = c$$

Find the equation of the circle (i.e., find a , b , and c) that passes through the three points (2, 1), (4, -1), and (1, 7). Plot the circle with the points. HINT: It is not possible to graph the circle with a function of the form $y = f(x)$ (why?); so it is easier to work with expressions using the `subs` command to obtain the necessary three equations. Start by assigning the above equation to the label `circle`. Substitute the three points into the circle to obtain three equations such as `eq1:=subs(x=2,y=1,circle);`. Then solve these three equations for a , b , and c and substitute them into the circle. Finally, plot the three points using `plot` and your circle using `implicitplot` and combine them using `display`. Don't forget you need `with(plots)`.

- Suppose a rubber band is stretched and the following data is recorded relating the restoring force to displacement.

Disp. (meters)	.02	.03	.05	.06	.09	.11	.12
Force (Newtons)	.10	.17	.28	.35	.52	.57	.65

.14	.16	.19	.22	.25	.27
.71	.77	.86	.98	1.04	1.10

Determine whether this data is best approximated using a linear, quadratic or cubic fit. In each case, plot the data against the curve.

5. The following table lists a measurement of the specific heat of air at various temperatures.

Temperature (K)	300	400	500	600
Sp. Heat (J/gm-K)	1.0045	1.0134	1.0296	1.0507

700	800	900
1.0743	1.0984	1.1212

Determine a linear curve that approximates this data, and plot the curve and the data in the same graph. In addition, compute the predicted values for specific heat (given by the linear approximation) corresponding to the temperature values in the table. Is there a relationship between the average of the predicted values and the average of the specific heat values in the table?

6. The following table lists a measurement of the specific heat of air at higher temperatures.

Temperature (K)	1000	1500	2000	3000
Sp. Heat (J/gm-K)	1.1910	1.2095	1.2520	1.2955

Determine a quadratic curve that approximates this data. Plot the curve and the data in the same graph. In addition, compute the predicted values for specific heat (given by the quadratic approximation) corresponding to the temperature values in the table. Is there a relationship between the average of the predicted values and the average of the specific heat values in the table?

7. Find the exact cubic curve passing through the data points of Exercise 6, and plot the cubic curve and the data in the same graph. Does the quadratic curve of Exercise 6 or the cubic curve of this exercise give a better description of the data? Be sure to consider extrapolating to higher and lower temperatures.
8. The following Maple command makes it easy to generate random data sequences of integers between -100 and 100 .

```
> r:=rand(-100..100):
```

For example, if you want to create random data sequences named **xs** and **ys** containing 10 data values each, simply do the following:

```
> xs:=[seq(r(),i=1..10)];
> ys:=[seq(r(),i=1..10)];
```

NOTE: The values obtained will be different each time you execute these commands.

Use Maple to create a random pair of data sequences named **xs** and **ys** containing 10 data values each. Find a linear curve that approximates this data and compute the predicted y values (given by the linear approximation) for the corresponding listed x values in **xs**. Compare the average

of the predicted y values to the given values in ys . Repeat this process 5 times. What do you conclude?

9. The ellipse $\frac{x^2}{9} + \frac{y^2}{16} = 1$ may be parametrized by $x = 3 \cos(t)$ and $y = 4 \sin(t)$. Check this by substituting the parametrization into the equation. Then, plot the ellipse using a parametric plot with constrained scaling.
10. Show that the parametric curve $x = 3 \cosh(t)$, $y = 4 \sinh(t)$ is part of the hyperbola $\frac{x^2}{9} - \frac{y^2}{16} = 1$. Then plot the piece for $-2 \leq t \leq 2$. Recall that

$$\cosh(t) = \frac{e^t + e^{-t}}{2}, \quad \sinh(t) = \frac{e^t - e^{-t}}{2} \quad \text{and} \quad \cosh^2(t) - \sinh^2(t) = 1$$

11. For each function, find the inverse function and plot the function and its inverse in the same plot on the given interval with the inverse thicker.

(a) $f(x) = \sqrt{144 - 9x^2}$ for $0 \leq x \leq 12$

(b) $f(x) = \sqrt[3]{1000 - 125x^3}$ for $0 \leq x \leq 10$

(c) $f(x) = x e^x$ for $-1 \leq x \leq 1$ NOTE: e^x is entered as `exp(x)`. Also LambertW is defined as precisely this inverse function.

12. For each function, plot the function and its inverse in the same plot with the inverse thicker.

(a) $y = x + \cos(x)$ for $-2\pi \leq x \leq 2\pi$

(b) $y = x^3 + 16x$ for $-2 \leq x \leq 2$

(c) $x = t^3 + 2t$, $y = t^3 + 3t$ for $0 \leq t \leq 2$

13. Graph each of the following equations by hand and then check your answers with a Maple plot. You may need to restrict the `view` on (a).

(a) $r = \frac{4}{1 + \cos(\theta)}$

(b) $r = 1 + 3 \cos(\theta)$

(c) $r = 4 \sin(3\theta)$

(d) $r = 3 \cos(4\theta)$

(e) $r^2 = 1 + \sin^2(\theta)$

(f) $r = \theta$, for $\frac{-3\pi}{2} \leq \theta \leq \frac{3\pi}{2}$ (Happy Valentine's Day)